

Sage 4.3 Calculus Commands

Although our initial Introduction to *Sage* included discussion of a number of *Sage* functions related to calculus, you may not have noticed that fact if you didn't look carefully at what *Sage* can do. This handout is, in the first instance, just a summary of *Sage's* calculus-related functionality. As always, I'm only scratching the surface. The *Sage* help command (say **derivative?** to get help on **derivative**, for instance) offers a lot more information on any of the commands used in this Worksheet.

Basic Commands

Limits

Sage's command for limits is, logically enough, **limit**. Notice how *Sage* handles the limits below, distinguishing among infinite limits, undefined limits, and indeterminate limits for functions that are still locally bounded. Notice also that unlike earlier versions of *Sage*, the current version gets one of these limits wrong.

```
var('x')
limit(sin(x)/x, x=0)
```

1

```
limit(1/x, x=0)
```

$+\infty$

```
limit(1/x^2, x=0)
```

$+\infty$

```
limit(x^2*sin(x), x=infinity)
```

und

```
limit(sin(1/x), x=0)
```

ind

Our investigation of derivatives could have been speeded up substantially had we been willing to use *Sage* uncritically as a black box for taking limits. Here, for instance, are calculations of the derivatives of the sine, exponential, and cube root functions. It would be useful review to try to remember how we obtained these results in class. Are you sure *Sage* doesn't have bugs that affect these calculations?

These calculation for the exponential function shows a pretty serious annoyance in *Sage*: It needs to know whether x is an integer, even though the result of the calculation doesn't depend on the answer to this questions. We therefore end up having to use *Sage's* **assume()** function to give both possible answers for the type of x .

```
var('h')
limit((sin(x+h) - sin(x))/h, h=0)
```

$\cos(x)$

```
limit((exp(x+h) - exp(x))/h, h=0)
```

```
Traceback (click to the left of this block for traceback)
```

```
...
```

```
Is x an integer?
```

```
assume(x, 'integer')
```

```
limit((exp(x+h) - exp(x))/h, h=0)
```

```
 $e^x$ 
```

```
forget()
```

```
assume(x, 'noninteger')
```

```
limit((exp(x+h) - exp(x))/h, h=0)
```

```
 $e^x$ 
```

```
forget()
```

```
var('a')
```

```
limit((x^(1/3) - a^(1/3))/(x-a), x=a)
```

```
 $\frac{1}{3} \frac{1}{a^{2/3}}$ 
```

Here's one last limit, just for fun. You might think about how to do it by hand.

```
limit(x^x, x=0)
```

```
1
```

Derivatives

Sage can also compute derivatives, either of expressions or of functions. **derivative()**, **differentiate()**, and **diff()** are three different names for the same function (or method) for doing this.

```
nestrاد = sqrt(x+sqrt(x+sqrt(x)))
```

```
nestrاد
```

```
 $\sqrt{x + \sqrt{x + \sqrt{x}}}$ 
```

```
diff(nestrاد, x)
```

```
 $\frac{1}{8} \frac{\left(\frac{\frac{1}{\sqrt{x}}+2}{\sqrt{x+\sqrt{x}}}+4\right)}{\sqrt{x+\sqrt{x}}\sqrt{x+\sqrt{x}}}$ 
```

```
_.simplify_rational()
```

```
 $\frac{1}{8} \frac{(4\sqrt{x+\sqrt{x}}\sqrt{x}+2\sqrt{x}+1)}{\sqrt{x+\sqrt{x}}\sqrt{x+\sqrt{x}}\sqrt{x}}$ 
```

```
f(x)=x^x
```


```
f
```

$$x \mapsto x^x$$

```
fprime = diff(f)  
fprime
```

$$x \mapsto (\log(x) + 1)x^x$$

```
pf = plot(f, (x,0,2), color='red', linestyle='-')  
pfprime = plot(fprime, (x,0,2), color='blue', linestyle='--')  
show(pf+pfprime,aspect_ratio=1, ymin=-2, ymax=2)
```



```
diff(sin(x),x)
```

$$\cos(x)$$

```
diff(sin(x),x,4)
```

$$\sin(x)$$

```
sin(x).diff(x,21)
```

$$\cos(x)$$

```
f(x)=sin(x)
```

```
diff(f)
```

$$x \mapsto \cos(x)$$

```
diff(f,x,4)
```

```
diff(f,x,4)(pi/4)  
 $x \mapsto \sin(x)$ 
```

$$\frac{1}{2} \sqrt{2}$$

Sage can even handle differentiation rules for arbitrary functions, like this:

```
function('f g h')
var('x')
```

x

```
diff(f(x)*g(x),x)
```

$$f(x) D[0](g)(x) + g(x) D[0](f)(x)$$

```
diff(f(x)/g(x), x)
```

$$-\frac{f(x)D[0](g)(x)}{g(x)^2} + \frac{D[0](f)(x)}{g(x)}$$

```
diff(f(x)*g(x)*h(x),x)
```

$$f(x) g(x) D[0](h)(x) + f(x) h(x) D[0](g)(x) + g(x) h(x) D[0](f)(x)$$

```
diff(f(g(x)),x)
```

$$D[0](f)(g(x)) D[0](g)(x)$$

```
diff(sqrt(f(x)),x)
```

$$\frac{1}{2} \frac{D[0](f)(x)}{\sqrt{f(x)}}$$

Integrals

Sage can also compute integrals, naturally using the function (or method) **integral()**.

```
integral(sin(x), x,0,pi)
```

2

```
(1/x).integrate(x,1,e)
```

1

```
integral(x/(1+x^2),x)
```

$$\frac{1}{2} \log(x^2 + 1)$$

```
integral(x/(1+x^4),x)
```

$$\frac{1}{2} \arctan(x^2)$$

Big deal so far, but here's one I bet you can't do by hand, $\int \frac{x}{1+x^3} dx$.

```
integral(x/(1+x^3),x)
```

$$\frac{1}{3}\sqrt{3}\arctan\left(\frac{1}{3}(2x-1)\sqrt{3}\right) - \frac{1}{3}\log(x+1) + \frac{1}{6}\log(x^2-x+1)$$

integral(x^3*e^x,x)

$$(x^3 - 3x^2 + 6x - 6)e^x$$

var('n')

integral(x^n,x)

Traceback (click to the left of this block for traceback)

...

Is n+1 zero or nonzero?

Ah, this is good! *Sage* should ask whether or not $n = -1$. After all, it makes a difference:

assume(n != -1)

integral(x^n,x)

$$\frac{x^{(n+1)}}{(n+1)}$$

forget()

integral(x^(-1),x)

$$\log(x)$$

Sums

In previous versions of *Sage*, sums were done with slightly unusual notation. **[1..10]** means the list **[1,2,3,4,5,6,7,8,9,10]**. For convenience, **range(n)** is the list of integers x such that $0 \leq x < n$. (Watch the endpoints here!)

range(5)

$$[0, 1, 2, 3, 4]$$

sum(k for k in [1..10])

$$55$$

Now, *Sage* has a slightly more familiar notation for sums, which even supports infinite sums! Here's how we might compute the sums

$$\sum_{k=1}^n k, \quad \sum_{k=1}^n k^2, \quad \sum_{k=1}^n k^3, \quad \sum_{k=1}^{\infty} \frac{1}{k^2}, \quad \sum_{k=1}^{\infty} \frac{1}{k^4}, \quad \text{and} \quad \sum_{k=1}^{\infty} \frac{1}{k^3}.$$

var('k n')

sum(k, k, 1, n)

$$\frac{1}{2}n^2 + \frac{1}{2}n$$

factor(_)

$$\frac{1}{2}(n+1)n$$

sum(k^2, k, 1, n)

$$\frac{1}{3}n^3 + \frac{1}{2}n^2 + \frac{1}{6}n$$

factor(_)

$$\frac{1}{6}(n+1)(2n+1)n$$

sum(k^3, k, 1, n).factor()

$$\frac{1}{4}(n+1)^2n^2$$

sum(1/k^2, k, 1, infinity)

$$\frac{1}{6}\pi^2$$

_.n()

1.64493406684823

sum(1/k^4, k, 1, infinity)

$$\frac{1}{90}\pi^4$$

sum(1/k^3, k, 1, infinity)

$$\zeta(3)$$

zeta(3).n()

1.20205690315959

Sage can compute integrals using Riemann sums. The right hand Riemann sum for computing the integral

$$\int_0^x t^3 dt$$

is

$$\sum_{k=1}^n \left(\frac{kx}{n}\right)^3 \cdot \frac{x}{n},$$

which *Sage* can write and evaluate.

sum((k*x/n)^3 * x/n, k, 1, n)

$$\frac{1}{4} \frac{(n^2+2n+1)x^4}{n^2}$$

We now want to take the limit of this as $n \rightarrow \infty$ in order to get the integral, right?

limit(_, n=infinity)

$$\frac{1}{4}x^4$$

Bingo! So


$$\int_0^x t^3 dt = \frac{x^4}{4},$$

just as we would have suspected.

More Complicated Tools

As one learns to work with *Sage*, one finds more and more ways to visualize and to work with the ideas of calculus. Here, for instance, is a little interactive tool for illustrating the idea of the derivative of a function as the slope of the tangent line. The blue curve is the function $f(x)$, the dotted black curve is its derivative $f'(x)$. Moving the slider moves a point along the curve and the tangent line at that point. As the tangent line moves, the green dot on the derivative curve traces out its slope.

```
var('x')
f(x) = sin(x)*e^(-x)
fp = f.diff()
p = plot(f,-1,5, thickness=2)
pp = plot(fp, -1, 5, color='black', linestyle=':')
pt_list = (-0.4+0.1*j for j in range(45))
@interact
def _(pt=pt_list):
    dot = point((pt,f(pt)),pointsize=80,rgbcolor='red')
    dotp = point((pt, fp(pt)), pointsize=80, rgbcolor='green')
    slope = fp(pt)
    sp = plot(f(pt)+slope*(x-pt),(x,-1, 5), color='green', thickness=2)
    html('<font color=red>Tangent to y = exp(-x)sin(x) at x = %s</font>%pt)
    show(dot + dotp + p + pp + sp, ymin = -.5, ymax = 1)
```



And here are two functions for working with Riemann sums. **right_riemann_plot** plots a curve and its right-hand Riemann sum, and **right_riemann_value** gives the value of this sum.

```
def right_riemann_plot(f, a, b, n):
    p = plot(f, a, b, thickness=2)
    deltax = (b-a)/n
    rectops = Piecewise([[(a+j*deltax, a+(j+1)*deltax),f(a+(j+1)*deltax)] for j in range(n)])
```

```

prectops = plot(rectops, color='green', fill='axis', fillcolor='green')
outline = []
for j in range(n):
    outline = outline + [(a+j*deltax,0), (a+j*deltax,f(a+(j+1)*deltax)), (a+(j+1)*deltax,f(a+(j+1)*deltax)), (a+(j+1)*deltax, 0)]
poutline = plot(line(outline, rgbcolor='black'))
show(p + prectops + poutline)


```

```

def right_riemann_value(f, a, b, n):
    deltax = (b-a)/n
    return(sum(f(a+(j+1)*deltax)*deltax for j in range(n)))

```

```
right_riemann_plot(sin,0,pi,4)
```




```
right_riemann_value(sin,0,pi,4)
```

$$\frac{1}{4} \pi + \frac{1}{4} \pi \sqrt{2}$$

```
right_riemann_value(sin,0,pi,4).n()
```

1.89611889793704

```
right_riemann_plot(sin,0,pi,25)
```



```
right_riemann_value(sin,0,pi,25).n()
```

```
1.99736741254563
```

```
integral(sin(x),x,0,pi)
```

```
2
```

Finally, here's a function that takes as variables a function f , endpoints a and b , and a number of slices n and returns the formal Riemann sum. At least for simple functions, we end up with expressions where *Sage* can take the limit to evaluate the integral.

```
def right_riemann_sum(f,a,b,n):
    deltax = (b-a)/n
    return sum(f(a + sage.calculus.calculus.var('k')*deltax) * deltax, k, 1, n)
```

```
right_riemann_sum(exp,0,1,3)
```

```
 $\frac{1}{3} e^{\frac{1}{3}} + \frac{1}{3} e^{\frac{2}{3}} + \frac{1}{3} e$ 
```

```
right_riemann_sum(exp,0,1,3).n()
```

```
2.02054276486660
```

```
right_riemann_sum(sin,0,pi,4)
```

```
 $\frac{1}{4} (\sqrt{2} + 1) \pi$ 
```

```
f(x)=x^2
```

```
right_riemann_sum(f,0,1,5)
```

```
 $\frac{11}{25}$ 
```

```
var('n')
```

```
rs = right_riemann_sum(f,0,1,n)
```

```
rs
```

```
limit(rs,n=infinity)
 $\frac{1}{6} (2n^2 + 3n + 1)$ 
 $\frac{1}{6} n^2$ 
```

$$\frac{1}{3}$$

integral(f(x),x,0,1)

$$\frac{1}{3}$$

var('z')

rs = right_riemann_sum(f,1,z,n)

rs

$$\frac{1}{6} \frac{((2n^2+3n+1)z^3 - 3(n+1)z^2 - 3(n-1)z - 2n^2 + 3n - 1)}{n^2}$$

limit(rs, n=infinity)

$$\frac{1}{3} z^3 - \frac{1}{3}$$

integral(f(x),x,1,z)

$$\frac{1}{3} z^3 - \frac{1}{3}$$

rs = right_riemann_sum(exp,0,z,n)

rs

$$\frac{(ze^z - z)e^{\frac{z}{n}}}{(ne^{\frac{z}{n}} - n)}$$

limit(rs, n=infinity)

$$e^z - 1$$

integral(e^x,x,0,z)

$$e^z - 1$$

Some Calculus Problems

As far as I can tell at this point, *Sage* doesn't have inert integrals like *Maple*, so some of the exercises in the *Maple* version of this worksheet which involve doing explicit changes of variable in inert integrals aren't so easy in *Sage*. Others of our exercises from the *Maple* version are easy. For instance, we can explicitly explore Taylor polynomials for the sine function by equating derivatives of $\sin x$ with derivatives of a generic polynomial at $x = 0$, and we can then interactively plot the first few Taylor polynomials.

var('a0 a1 a2 a3 a4 a5')

f(x) = a2*x^2 + a1*x + a0

s(x)=sin(x)

solve([f(0)==s(0),diff(f,x)(0)==diff(s,x)(0),diff(f,x,2)(0)==diff(s,x,2)(0)], a0,a1,a2)

$$[[a_0 = 0, a_1 = 1, a_2 = 0]]$$

f(x) = a3*x^3 + a2*x^2 + a1*x + a0

solve([f(0)==s(0),diff(f,x)(0)==diff(s,x)(0),diff(f,x,2)(0)==diff(s,x,2)(0),diff(f,x,3)(0)==diff(s,x,3)(0)], a0,a1,a2,a3)

$$[[a_0 = 0, a_1 = 1, a_2 = 0, a_3 = (-\frac{1}{6})]]$$

f(x) = a5*x^5 + a4*x^4 + a3*x^3 + a2*x^2 + a1*x + a0

```
solve(list(diff(f,x,k)(0)==diff(s,x,k)(0) for k in range(6)), a0,a1,a2,a3,a4,a5)
```

$$[[a_0 = 0, a_1 = 1, a_2 = 0, a_3 = (-\frac{1}{6}), a_4 = 0, a_5 = (\frac{1}{120})]]$$

```
@interact
```

```
def_(terms=(j for j in range(17))):  
    x = sage.calculus.calculus.var('x')  
    sinplot = plot(sin(x), x,-18, 18, rgbcolor='blue')  
    taypoly = sum((-1)^k*x^(2*k+1)/(2*k+1).factorial() for k in range(terms+1))  
    tayplot = plot(taypoly, x, -15,15, rgbcolor='red')  
    html('<font color=red>y = sin(x) and its Maclaurin series of degree %s</font>'%(2*terms+1))  
    show(sinplot + tayplot, xmin=-18, xmax=18, ymin = -2, ymax = 2)
```

terms

y = sin(x) and its Maclaurin series of degree 5

